# How to make traditional output for derivatives

Hello again after some pause. i have a problem how to present partial derivatives in traditional form, not as Mathematica gave it to me in its notation. So I want to present this

Subscript[S, 1]* (F1^(4,0))[x,y]+Subscript[S, 2]* (F1^(2,2))[x,y]


to have a look $S_1\frac{\partial ^4F_1(x,y)}{\partial x^4}+S_2\frac{\partial ^4F_1(x,y)}{\partial x^2\partial y^2}$

And if I have for example lot of different combination of derivatives terms I want to do this automatically . Is it possible?

• I posted a solution to this problem as part of my answer here. Its advantage is that it is more compact and that it automatically uses straight derivative symbols (instead of curly ones) when you're not doing partial derivatives.
– Jens
Apr 20 '13 at 16:13

The problem with using TraditionalForm@Defer is that it won't work as soon as the Defer is gone. So you always need an additional wrapper, different from the simple TraditionalForm wrapper, to get the desired output. It can sometimes be desirable to have the derivative formatted automatically for all TraditionalForm environments, e.g., in Graphics labels etc.

If, as you mention in the question, you have numerous combinations of derivatives, then the output quickly becomes cluttered if you keep writing out all the function arguments in a partial derivative. This is why I formatted the derivative without arguments in this answer.

An additional formatting requirement in traditional form would be to write ordinary derivatives with a straight derivative symbol to distinguish them from partial derivatives. This isn't automatically done by Mathematica in TraditionalForm output, so I added that case distinction in the linked answer.

The downside of the shortened notation in that answer was that you can't copy the formatted output and re-use it as input by pasting it back into a new line.

Here is a way to get the advantages of more readable short-hand notation in the displayed output while at the same time maintaining the ability to evaluate the output later:

Derivative /:
MakeBoxes[Derivative[α__][f1_][vars__?AtomQ],
TraditionalForm] := Module[{bb, dd, sp},
MakeBoxes[dd, _] ^=
If[Length[{α}] == 1, "\[DifferentialD]", "∂"];
MakeBoxes[sp, _] ^= "\[ThinSpace]";
bb /: MakeBoxes[bb[x__], _] := RowBox[Map[ToBoxes[#] &, {x}]];
TemplateBox[{ToBoxes[bb[dd^Plus[α], f1]],
ToBoxes[Apply[bb,
Riffle[Map[
bb[dd, #] &, (Pick[{vars}, #]^Pick[{α}, #] &[
Thread[{α} > 0]])], sp]]],
ToBoxes[Derivative[α][f1][vars]]}, "ShortFraction",
DisplayFunction :> (FractionBox[#1, #2] &),
InterpretationFunction :> (#3 &), Tooltip -> Automatic]]



$\frac{d f}{d x}$

TraditionalForm[D[f[x, y], x, x]]


$\frac{\partial^2 f}{\partial x^2}$

Note the absence of arguments $(x)$ and $(x,y)$, respectively, and the different symbols for partial and ordinary derivatives. Also, you can now copy any of the above outputs and paste them into an Input cell. The result is again recognized as the original derivative, without loss of information.

What I added to the original solution linked above is a TemplateBox that specifies both a DisplayFunction (using the original formatting in the previous answer), and an IntepretationFunction which simply contains the box form of the original derivative expression. The latter is used when you try to evaluate the output. Since the function arguments are kept in that expression, it can be evaluated without problems.

Edit:

Originally I allowed this simplified derivative display only to be applied when al the function variables are symbols, as mentioned in the comments. The reason is that the simplified notation is clumsy when the chain rule needs to be applied. In that case, it's better to identify the function slots as it's done in the built-in display of Derivative.

On the other hand, the restriction to symbolic variables may be too strong if you also want to get the shortened display with functions that have one or more variables set equal to a constant. So I now allow the display to work as long as the arguments are atomic, not necessarily symbols. This makes it possible to get the following (note x = 0):

TraditionalForm[D[f[0, y], y]]


$\frac{\partial f}{\partial y}$

The output can still be copied into an input cell to get back the original information: f^(0,1)[0,y].

Here I verify that for compositions of functions, I still get the more verbose display I prefer:

TraditionalForm[D[f[g[x], y], x]]


$f^{(1,0)}(g(x), y)\,\frac{d g}{d x}$

• Defer is stripped on input, but if you don't want that why not use HoldForm instead? I don't think I understand where you're going with this answer. (That doesn't mean it's wrong!) Sep 4 '13 at 2:34
• Maybe this blog entry which is linked in the earlier answers I linked to will answer this. Defer is discussed as an alternative there, too.
– Jens
Sep 4 '13 at 2:58
• I find this very useful. :) May I ask how do you handle arbitrary dimensions? When I have to do this I'm using D[t[##], #] & @@ Array[x, 3] where 3 can be replaced. But x[i] is not a symbol, so I have to delete __Symbol from your code. Then it is ok, but the question is, since you've given this constraint I'm assuming you handle this in different way, is there better than Array?
– Kuba
Nov 28 '13 at 14:28
• @Kuba The reason I put the Symbol constraint in the definition is that I didn't want this short notation to be used when the chain rule is needed. In that case, I want to retain the clearer formal notation to avoid confusion. So for example TraditionalForm[D[f[g[x], y], x]] will be displayed in the standard way. By the way, the chain rule wouldn't even be use at all if you use the other method, TraditionalForm@Defer[D[f[g[x], y], x]]. But this may be a matter of taste and depends on your needs. You could also use Symbolize on your variables and leave the __Symbol constraint in.
– Jens
Nov 28 '13 at 21:13
• Thanks! This code works great :) it would be helpful if you could show how to adapt it to display the even more simplified form of $f_x, f_{xy}, ...$ etc instead of $\partial f / \partial x, \partial^2 f / \partial x \partial y, ...$ etc Dec 11 '17 at 12:29

Besides of TraditionalForm you should use Defer to stop automatic evaluation of the underlying expression, so it should look like this :

TraditionalForm @ Defer[   Subscript[S, 1] D[F1[x, y], {x, 4}]
+ Subscript[S, 2]*D[F1[x, y], {x, 2}, {y, 2}]]


In order to change the order of terms one could use also undocumented
PolynomialForm[..., TraditionalOrder -> True]:

PolynomialForm[
Defer[ Subscript[S, 1] D[Subscript[F, 1][x, y], {x, 4}] +
Subscript[S, 2] D[Subscript[F, 1][x, y], {x, 2}, {y, 2}]],